THE SAS/GRAPH® ANNOTATE FACILITY

Anthony L. Friebel
SAS Institute Inc.

INTRODUCTION

ABSTRACT

The annotate facility permits you to customize SAS/GRAPH procedure output. The facility is capable of defining polygons, controlling pen movement, placing text on the page, and so on. It adds to existing procedures the capability of producing custom graphic output from a DATA step.

THE ANNOTATE FACILITY

Welcome to the annotate facility, a command driven graphics customization tool. The commands, or functions, are contained in a SAS data set. This data set is then specified in the ANNOTATE= option of the graphics procedure.

PROC GMAP DATA = sample.data
   MAP = sample.map
   ANNOTATE = company.logo;
ID id;
CHORO id / discrete
   nolegend
   ANNOTATE = map.detail;
RUN;

The ANNOTATE= option is available in both the PROC statement and in interior statements which produce graphics as the above sample indicates. Both pass instructions to the annotate facility, but the PROC statement specification is special. When the ANNOTATE= option is specified in the PROC statement the data set named will be applied globally to all interior graphics-producing statements within the procedure range. This is useful in applications where a company logo is to be placed on each plot generated. The statement specification is active only for the statement which invoked the ANNOTATE= option. Its output is placed on the graph in addition to any annotate output generated from the PROC statement data set.

Though only one annotate data set may be specified per graphics statement, each of multiple statements may contain a separate annotate data set.

Annotate data sets are characterized as having reserved variable names for conveying information. Each observation is a complete command. Observations may be designated as occurring BEFORE or AFTER any graphic output is produced by the procedure. This division of processing permits you to place items on the graph in relation to the order of graphics generation. If labelling PROC GPLOT area fills, for example, you will want to place the text 'AFTER' the area is drawn on the graph. Otherwise, the text will be covered up when the area is filled in.

The immediate uses of the annotate facility center around accurate data dependent positioning of text labels, particularly in mapping applications. Annotate uses may also include:

- Placement of city names on maps
- Identification of areas
- Highlighting data minimum, maximum, or inflection points
- Special presentation graphics

Because annotate commands are specified in a SAS data set, the facility is flexible enough that an entire graphics procedure may be simulated with DATA step coding mechanisms.

SUPPORTED FUNCTIONS

The annotate facility is directed by a series of observations and variables contained in a SAS data set. Observations are the commands, variables are the qualifiers of information in the command. The variables have special reserved names which are indicative of their use.

<table>
<thead>
<tr>
<th>Variable</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>X</td>
<td>The X or horizontal coordinate</td>
</tr>
<tr>
<td>Y</td>
<td>The Y or vertical coordinate</td>
</tr>
<tr>
<td>FUNCTION</td>
<td>Particular feature you want to accomplish</td>
</tr>
<tr>
<td>other variables</td>
<td>Additional information</td>
</tr>
</tbody>
</table>

X and Y

X and Y specify the coordinates on the graph where the FUNCTION will be applied. They may be considered in any of a number of available reference systems. These coordinate systems translate/transform input values to the appropriate screen locations.

The reference systems may be used in any combination to specify screen locations in the annotate data set. The (X, Y) coordinate pair is internally transformed by the annotate facility to locate the requested location to its formal screen designation. The X and Y variables need not be referenced with the same system value.

FUNCTION

FUNCTION is a character variable whose value indicates which annotate feature you want
performed. Below is a list of some of the available annotate functions and a short description of each.

MOVE
Permits movement to a specific point on the graphics page without drawing a line. MOVE is most often used to prepare for a DRAW command.

DRAW
Used to draw a line on the graphics page. You have control over color, style and thickness of the drawn line.

LABEL
Places text on the graphics page. You may specify the color, size, font, base angle and rotation of the characters displayed.

OTHER VARIABLES
Additional variables may be supplied to most functions. Color, line weight, line style, font or pattern are some of the descriptive qualifiers that may be used.

DEMONSTRATION
To illustrate use of the annotate facility, suppose we want to

1) draw a box on the screen
2) put the word 'ANNOTATE' in the box.

We must decide how we are to accomplish our task. We will use the MOVE and DRAW functions to create the box. After the box is drawn, we will use the LABEL function to place the text string inside it.

First, we create the annotate data set.

```
DATA BOX;
LENGTH FUNCTION $ 8.;
FUNCTION= 'MOVE'; X= 30; Y= 10; OUTPUT;
FUNCTION= 'DRAW'; X= 50; Y= 10; OUTPUT;
FUNCTION= 'DRAW'; X= 50; Y= 30; OUTPUT;
FUNCTION= 'DRAW'; X= 30; Y= 30; OUTPUT;
FUNCTION= 'DRAW'; X= 30; Y= 10; OUTPUT;
FUNCTION= 'LABEL';
TEXT= 'ANNOTATE'; X= 40; Y= 20; OUTPUT;
RUN;
```

Next we specify this data set to the graphics procedure using the ANNOTATE= option.

```
PROC GANNO ANNOTATE=BOX;
RUN;
```

This code produces the following output:

```
ANNOTATE
```

In our next example, we want to place the city name "Reno" onto a map of Nevada. The map data library contains a city database called USCITY. It contains the geographic locations of many United States cities, as well as some additional demographic information. We will use this data set to supply the information on Reno.

We create our annotate dataset with the USCITY information on Reno.

```
DATA RENO;
/* request data dependent placement */
XSYS= '2'; YSYS= '2';
/* city names are upper/lower case */
SET MAPS.USCITY;
IF UPCASE(CITY) = 'RENO';
/* string will be 'Reno' */
POSITION= '0';
TEXT= ' RENO' || CITY;
OUTPUT;
RUN;
```

Next, we obtain the Nevada state outline from the US map data set.

```
DATA NVMAP;
SET MAPS.US;
IF STATE=STFIPS('NV');
RUN;
```

Then specify the RENO dataset in the ANNOTATE= option of PROC GMAP’s CHORO statement.
TITLE J=L 'LOCATION OF RENO, NEVADA';
PROC GMAP DATA = NVMAP
    MAP = NVMAP;
    ID STATE;
CHORO STATE / NOLEGEND
    ANNOTATE = RENO;
    PATTERN C=GREEN V=E R=199;
RUN;

This code produces the following output:

LOCATION OF RENO, NEVADA

SUMMARY

The annotate facility is a flexible means for adding descriptive annotation to SAS/GRAPH procedure output. Using the power of the SAS DATA step and SAS macro facility, it can be extended to perform a wide range of graphics customization.

SAS and SAS/GRAPH are registered trademarks of SAS Institute Inc, Cary, North Carolina, USA.